
Last ADC12 top-box for the micro1401

We currently have one remaining ADC12 top-box available for a micro1401 mk | unit.
This will expand a micro1401 unit with 12 additional waveform input channels. Anyone
interested in expanding their micro1401 with the last ADC12 top-box should contact
sales@ced.co.uk.

micro1401 mk | with ADC12 top-box
The ADC12 top-box will still be available for Micro1401 mk Il and Micro1401-3 units.

Meetings and events

ANS Sensorimotor Control Satellite Meeting
Prince Of Wales Medical Research Institute,
Sydney, Australia

January 30" — 31 2010

Australian Neuroscience Society/Australian Physiological Society Joint Meeting
Darling Harbour Convention Centre,

Sydney, Australia

January 31% — February 3 2010

Latest versions of Spike2 and Signal

Updates for Spike2 and Signal are available from the CED downloads page, or by
clicking on the links in the table below. Demonstration versions of the latest software are
also available.

Spike2 downloads Signal downloads
Spike2 version 7.02 Signal version 4.06
Spike2 version 6.13 Signal version 3.12
Spike2 demo Signal demo



http://www.powmri.edu.au/ans2010motorcontrol/
http://www.sallyjayconferences.com.au/sydney2010/
http://www.ced.co.uk/uplsu.shtml
http://www.ced.co.uk/s2wu7u.shtml
http://www.ced.co.uk/s2wu6u.shtml
http://www.ced.co.uk/s2demu.shtml
http://www.ced.co.uk/sigw4u.shtml
http://www.ced.co.uk/sigw3u.shtml
http://www.ced.co.uk/sigdemu.shtml
mailto:sales@ced.co.uk

| am using the graphical editor in Spike2 to generate 5 different stimulus
protocols, stored in key sections A-E, which repeat a set number of times each
and play out one after another. | would like the whole sequence to repeat up to
15 times before stopping.

The graphical editor can make use of variables and branch statements for flow-
of-control between individual sections in the sequence. For the question above,
the use of a variable that acts as a counter along with a branch statement that
checks the variable value could be used to repeat the whole sequence a given
number of times.

First, create a variable in the Initial section of the graphical editor by clicking

and dragging the 1% symbol from the palette to the blue control track. Set the
variable Value field to the number of repeats required to act as a counter.

Initial section of the graphical editor

Next, select Key F from the Current Section drop-down list (this is the next
‘free’ section after the defined pulse protocols held in A-E) and place another
variable on the control track. Select Sub value from the drop-down list, set
Target var to 1 and Value to 1. This will subtract 1 from our variable counter
each time the sequence has run.

Finally, place a branch command after the variable, by clicking and dragging the

0 symbol to the control track. Select Variable above from the drop-down list,
set the Variable field to 1 and the Value field to 0. Then set the branch
destination to Key A. Each time the sequence reaches the end of this section
the counter variable is decreased by 1 and checked by the branch statement. If
the variable is above 0, the sequence repeats, otherwise it proceeds to the end
of the last section where it halts. The attached sampling configuration file,
SeqRepeat.S2C, contains the above example in the graphical editor.

Back to top

Right-click the sampling
configuration icon and
save to disk.

SeqRepeat.S2C

CED
You can duplicate pulses and control statements in the current section of the graphical editor by clicking on the item and dragging it to a new position while holding down Ctrl

Signal

Q. Is there any way to add channel modifications, such as rectify and smooth, to a ~ Backto to
copy of an existing channel so that the original data is not permanently modified?

A. Virtual channels can be used to carry out a wide variety of channel arithmetic (for
example, sums and differences of channels), generate new waveforms based on
sine, square or triangle envelopes, and add channel processes such as DC offset
removal, rectification and smoothing to copies of raw data channels both on-line
and off-line. Virtual channels are created from the Analysis menu > Virtual
channel command. The new virtual channel requires a sampling rate and can be
matched to an existing waveform channel using the drop-down list or you can
specify a sample interval and align time.

There are a number of expression functions that can be typed into the e
Expression field of the dialog to generate virtual channels (a full list of these can
be found in the on-line help) or you can build expressions using the >] button

and selecting the operation to perform from the menu. For this example, select
Waveform from channel from the menu and select the channel you want to
process from the resultant dialog. This creates the expression Ch(n), where Ch()
is the function to copy a waveform channel and n is the channel number. To add

a channel process to the channel, select the text Ch(n) in the main Virtual
channels dialog by clicking and dragging with the mouse. Next, click >] and
select Rectify from the Channel process functions sub-menu. This
creates the expression Abs(Ch(n)) in the dialog and generates the result of

this expression in the virtual channel displayed in the data file.

w;

Virtual channels for EMG Waveform.CFS

Virtual channel showing rectified EMG data from channel 1


CED
The sub menu for building expressions also holds a list of the previous expressions used


Channel process functions can be applied in sequence. For example, to add a 3
point smooth to the rectified data simply select the text Abs(Ch(n)) in the
dialog and select the 3 point smooth option from the Channel process
functions menu.

Scripts: Spike2 E

Q. It would be very useful to be able to create your own arbitrary waveforms for
output in Spike2 just using the mouse. Are there any plans to include this as an
option?

A. The attached script, DrawPlayWave.s2s, is an example script that allows the

drawing of an arbitrary waveform for output from the 1401 DAC'’s using the
mouse pointer. The user specifies the output rate, waveform duration and
associated key for the waveform before drawing the shape required in a result
view window by holding down shift together with the left mouse button and
dragging the mouse. The waveform can be added to the play wave area of the
sampling configuration by clicking the Add button in the script toolbar.

This script requires Spike2 version 7.01 or later.

Scripts: Signal

Q. | would Ii@ be able to export the number and times of occurrence of action
potentials in a Signal data file as text.

A. The attached script, APcount.sgs, is an example script that gets the time and
count of action potentials that cross a user defined threshold from nominated
frames in a Signal data file and prints the results to the Log window. The script
uses the ChanSearch() function to find all threshold crossings in each frame.

Did you know...?

The Power1401 and Power1401 mkll interfaces both have 16 waveform input channels
as standard. Of these, 8 are available as BNC inputs on the front panel with the further 8
available as pin inputs on the rear analogue expansion port. We supply patch panels and
break-out boxes that allow connection to the rear input ports via BNC connections if
required.

Recent questions

Q. I’'m using a mechanical switch to log events during a recording but | find that the
switch is generating multiple markers each time it is used. Is there any way to

remove these extra events so that | only get one each time the switch is activated?

A. There is a Debounce field in the channel parameters dialog for Event and Digital
Marker channels that will set the minimum acceptable interval (ms) between
consecutive events so that events closer than this to the previous event are not
saved to disk. This is useful when switches have bouncy contacts and generate
multiple switch closures in a short period of time.

Channel parameters

Channel |3 .v T¥pe |Event+ .v Title | Switch

1401 Port |0 (DigIn 8) ™ Maximum event rate | 50 Hz
Comment

Debounce {ms) | 0.01

lCanceIl [ 0K J

Help

Channel parameters dialog

If you have this problem with a Level event channel, you must fix it by using an
electronic switch debouncer as the Debounce option is not available in this case.

Back to top

Right-click the script icon
and save to disk.

If you have any problems
opening the embedded
scripts in this newsletter
please let us know.

Back to top

Right-click the script icon
and save to disk.

If you have any problems
opening the embedded
scripts in this newsletter
please let us know.

'&DrawPlayWave.s2s|Spike2 script to draw an arbitrary waveform for output from the 1401 DACs during sampling

'The user specifies the output rate, waveform duration and associated key, and is then presented with

'a result view in which they can draw the required waveform using the mouse. 

'Drawing takes place if the Shift key is pressed together with the left mouse button.

'Once the waveform is drawn, pressing the 'Add' button adds it to the sampling configuration play wave section



'The script is a 'work in progress' and is offered without guarantees.

'You must test it to see whether it is suitable for your application.



'CED 18/01/10



var plot%;                                              				'Result view for waveform creation

var pWaveRate := 1000;                                  				'Waveform output rate

var pWaveInt;                                           				'Sample interval (1/rate)

var pWaveLength := 0.5;                                 				'Waveform duration

var pWavePts%;                                          				'Number of data points

var pWaveKey$ := "a";                                   				'Key to trigger wave

var pWaveLabel$;                                        				'Label

var Dac% := 0;                                          				'DAC for output

var ok%;

var oldX%;                                              				'Previous mouse position (for interpolation) 

var oldY;

var outMax := 5;														'DAC full scale value

var outZero := 0;                                                       'DAC zero value

var wUnits$ := "Volts";                                                   'Units



if App(-1)<701 then Message("This script needs at least Spike2 version 7.01"); halt endif;	'Checks for the latest version of Spike2



ToolbarSet(1, "Quit");                                         			'Toolbar to control script

ToolbarSet(3, "New", MakeWave%);

ToolbarSet(4, "Add", AddPWave%);

Toolbar("Script to draw arbitrary waveforms using the mouse",1023);



Func MakeWave%()                                               			'Function to generate waveform 

DlgCreate("Playwave parameters");                              			'Parameters dialog

DlgReal(1,"Output frequency (Hz)",1,100000);

DlgReal(2,"Playwave duration (s)",0.01,1000);

DlgString(3,"Key",1);

DlgString(4, "Label",10);

DlgInteger(5, "DAC",0,3);

DlgReal(6, "DAC full scale value",0.0001,1000000);

DlgReal(7, "DAC zero value", 0, 1000);

DlgString(8, "Units",10);

ok% := DlgShow(pWaveRate,pWavelength,pWaveKey$,pWaveLabel$,Dac%,outMax,outZero,wUnits$);    	'Return variables

if ok% > 0 then

	if plot% > 0 then                                            		'Clear any existing result views

		View(plot%);

		FileClose(0,-1);

		plot% := 0;

	endif;

	pWaveInt := 1.0/pWaveRate;                                   		'Get sample interval

    pWavePts% := pWavelength*pWaveRate;                          		'Get number of points to output

    plot% :=SetResult(1,pWavePts%,pWaveInt,0,"PlayWave","s");    		'Set up result view with appropriate bin size etc

    DrawMode(1,2);                                                      'Draw in waveform mode

    YRange(1,outZero - (outMax-outZero),outMax);

	ChanUnits$(1,wUnits$);	

    Window(0,0,100,100);

    FrontView(plot%);

	ToolbarMouse(plot%,0,5,5,Down%,Up%,Move%);                   		'Set up mouse capture requiring shift and left mouse button to activate functions

    ToolbarText("Click 'Add' to add the waveform to the sampling configuration for output");		

endif;

return 1;

end;





Func Down%(vh%,chan%,x,y,flags%)                                 'Function for left mouse button press

if XtoBin(x) >= 0 and XToBin(x) <= Len([])-1 then                	'If within range

    [XToBin(x)] := y;                                            		'Set bin at mouse x position to mouse y position

    oldX% := XtoBin(x);                                          	'Update variables with last mouse position

    oldY := y;

endif;

return 1;

end;





Func Up%(vh%,chan%,x,y,flags%)                                   		'Function for left mouse button up



return 1;

end;



Func Move%(vh%,chan%,x,y,flags%)                                		'Function for mouse movement

var bin%;                                                       		'Current bin position of mouse

var pts%;                                                       		'Number of points moved since last time function called



bin% := XToBin(x);                                              		'Get bin position

if bin% >= 0 and bin% <= Len([])-1 then                         		'If within area

    if bin% <> oldX% then                                       		'Has it moved?

        pts% := Abs(bin%-oldX%);                                		'Count points moved since past call

        var line[pts%+1];                                       		'Declare array to contain interpolated points

            if bin% > oldX% then                                		'If mouse moved right (positive change)

                ArrConst(line[1:pts%],(y-oldY)/pts%);           		'Fill array with incremental y change per bin

                ArrIntgl(line[]);                               		'Integrate to get slope

                ArrAdd(line[],oldY);                            		'Add starting level to array

                ArrConst([oldX%:pts%],line[]);                  		'Add into result view

            else		

                ArrConst(line[1:pts%],(oldY-y)/pts%);           		'Same process for moving left (negative change)

                ArrIntgl(line[]);		

                ArrAdd(line[],y);		

                ArrConst([bin%:pts%],line[]);		

            endif;		

    else		

        [XToBin(x)] := y;                                       		'If not moved in x, update current bin with y mouse position

    endif;		

    oldX% := bin%;                                              		'Update variables with mouse position

    oldY := y;

endif;

return 1;

end;





Func AddPWave%();

PlayWaveAdd(pWaveKey$,pWaveLabel$,Dac%,pWaveRate,View(plot%,1).[]);  	'Add waveform section to playwave area in configuration

return 1;

end;



CED

'$APcount|Counts and gets times for all spikes that cross a threshold in selected frames



'The script is a 'work in progress' and is offered without guarantees.

'You must test it to see whether it is suitable for your application.



'CED 18/01/10



var count%;                'Count of spikes

var current;               'Current spike being counted

var chan%;                   'Spike channel

var gotSpikeFlag% := 0;      'Flag to show if any spikes found

var frameCounter%;         'Current frame no.

var data%;                  'Data file

var spikeTimes[1000];		'Array to hold actual times of spikes

var printCount%;           'Current printed value

var sFrame%,eFrame%;        'Start and end frames for analysis

var ok%;                   'ok flag for dialog



if ViewKind() = 0 then              'Check we have a data file open

data%:= View();                      'If so, use it

FrontView(data%);

else

data%:= FileOpen("",0);        'If not prompt user to select one

WindowVisible(1);

endif;

if data%<0 then Message("Unable to open new data file");Halt; endif;



sFrame% := 1;                  'Get default values for start and end frames

eFrame% := FrameCount();



View(LogHandle());           'Get the Log window

EditSelectAll();              

EditClear();                  'Clear any existing text

WindowVisible(0);

Window(20,20,80,80);

PrintLog("Frame\tSpikes\n");      'Print headings

View(data%);

DlgCreate("Select frames to measure");

DlgChan(1, "Spike channel",1);

DlgInteger(2, "Start frame",1,FrameCount());        'Settings 

DlgInteger(3, "End frame",1,FrameCount());

ok% := DlgShow(chan%,sFrame%,eFrame%);        'Get channel, start and end frames for processing

if ok% > 0 then

    HCursorDelete(-1);		'Delete existing horizontal cursors

    HCursorNew(chan%);				'Set horizontal cursor on channel 1

    Interact("Set horizontal cursor at threshold level and press OK",1023); 'Allow user to set level

    

    

    for frameCounter% := sFrame% to eFrame% do			'Loop through each frame

        count% := 0;											'Reset spike counter

        Frame(frameCounter%);								'Go to frame	

        current := MinTime();								'Set variable to beginning of frame

        repeat													'Loop through each crossing	

            gotSpikeFlag% := ChanSearch(chan%,7,current,MaxTime(),HCursor(1));

            if gotSpikeFlag% = 0 then							'If cursor has moved (spike detected)

                spikeTimes[count%] := current;

                count% += 1;									'Add 1 to spike count

            endif;													

            until gotSpikeFlag% < 0;							'Stop when no spike found		

        if count% > 0 then

            ArrMul(spikeTimes[],1000);						'Convert times to ms	

            PrintLog("%d\t%d",frameCounter%,count%);	    'Print frame number and spike count to log window

            for printCount% := 0 to count%-1 do

                PrintLog("\t%g",spikeTimes[printCount%]);        'Print spike times to log window

                next;

            PrintLog("\n");

        else

            PrintLog("%d\t%d\t\n",frameCounter%,count%);	'Print frame number and spike count to log window		

        endif;

        next;

    

    FrontView(LogHandle());         'Bring the Log view to the front

endif;

User group
Try the CED Forums bulletin board for software and hardware support
Back to top

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk.

